
Введение 
 

В данном выпуске собраны примеры решения основных статических 
задач сопротивления материалов на конечно-элементном комплексе 
ANSYS. Рассмотрены следующие разделы: расчет ферм на растяжение - 
сжатие, изгиб балок, расчет рам, задачи кручения, расчеты кривых стерж-
ней, балок с несимметричным или переменным поперечным сечением, ба-
лок на упругом основании, внецентренное растяжение - сжатие, учет на-
чальных и температурных деформаций. По каждому разделу приведены 
программы для ANSYS с подробными комментариями, составленные ав-
тором и решающие одну из задач раздела. Программы проверялись на 
комплексе ANSYS версии 5.3 , работающем под управлением операцион-
ных систем Windows95 и WindowsNT. Дано сравнение полученных резуль-
татов с известными аналитическими решениями. 

Все рассмотренные задачи относятся к «прямым» задачам сопротив-
ления материалов, в которых для заданных конструкций по известным си-
ловым факторам требуется рассчитать напряжения, реакции и т.п. Оптими-
зационные задачи сопротивления материалов, которые также можно ре-
шать на ANSYS, здесь не рассматриваются. 

Приведенные программы предназначены для выполнения в пакетном 
режиме BATCH. При использовании данных программ в интерактивном 
режиме, если предполагается их дальнейшая отладка без выхода из 
ANSYS, рекомендуется закомментировать первую строку /BATCH, послед-
нюю строку /EXIT и строку с командой /SHOW... При необходимости пере-
назначить графический вывод на экран (вместо вывода в файл) в операци-
онных системах Windows95 и WindowsNT можно по команде 

/SHOW,WIN32 
Задания желательно выполнять последовательно, поскольку матери-

ал каждого отдельного раздела опирается на предыдущее изложение. 
 
 
 
Автор с благодарностью примет замечания по данному выпуску, ко-

торые можно направить по e-mail: nasedkin@ns.math.rsu.ru. 

1. Расчет ферм 
 
 В качестве примера расчета фермы рассмотрим задачу № 4.67 (5.7) 
из задачника [1]. В данной задаче требуется определить опорные реакции 
и усилия в стержнях фермы, показанной на рис. 1.1 вместе с действующи-
ми в узлах силами.  

 
Рис. 1.1. 

 
 Отнесем ферму к декартовой системе координат Oxy (рис. 1.1), и по 
рисунку, приведенному в задачнике, рассчитаем координаты ее узлов. На 
рис. 1.1 узлы фермы обозначены метками n1, ..., n5. Эти узлы имеют сле-
дующие координаты (в м): n1 - (0;0), n2 - (2;0), n3 - (4;0), n4 - (2.5; 3 2/ ), 
n5 - (1; 3 ). 
 Как известно, ферма есть конструкция, состоящая из стержней, со-
единенных между собой шарнирами (узлами фермы). Поскольку силы, 
действующие на ферму, прилагаются в шарнирах, то все стержни фермы 
испытывают только осевые усилия растяжения - сжатия. Следовательно, 
подходящими КЭ для стержней ферм являются стержневые элементы 
LINK1 в случае плоских ферм и LINK8 в случае пространственных ферм. 
Узлы конечно-элементной модели будут тогда совпадать с узлами фермы, 
а каждый стержень будет отдельным КЭ. 

Приведенная на рис. 1.1 ферма с количеством стержней e=7 и коли-
чеством узлов n=5 является статически определимой, так как для нее вы-
полняется условие статической определимости плоских ферм: e=2n-3. Дан-
ную задачу можно решать методами теоретической механики, и тогда не 
существенны никакие определяющие параметры стержней, кроме их длин. 
Однако, для КЭ LINK1 (или LINK8) требуется задать по крайней мере одно 
материальное свойство (модуль Юнга EX) и одну константу КЭ (площадь 
поперечного сечения AREA). Примем произвольно, что для всех стержней 

стержней фермы EX = 2 10  (н / м11 2⋅ )  и AREA = ⋅ −1 10 4  (м). Отметим, что 
в рассматриваемой статически определимой задаче значения этих парамет-
ров не будут влиять на итоговые искомые величины, подлежащие опреде-
лению. 

Приведем листинг простой программы для ANSYS, предназначенной 
для решения описанной задачи в пакетном режиме BATCH. Строки про-
граммы здесь пронумерованы последовательно для удобства дальнейших 
комментариев. В компьютерном файле нумерация строк должна отсутст-
вовать! 
 
1   /BATCH 
2   /COM,                Файл SMs1.inp 
3   /COM,            Пример расчета плоской фермы 
4   /COM, (Мещерский И.В. Сборник задач по теоретической механике. 
5   /COM,  М.: Наука, 1986. Задача 4.67(5.7)) 
6   F_R='SMs1'       ! Имя файла для вывода результатов 
7   /PREP7 
8   ET,1,LINK1       ! Стержневой КЭ LINK1 
9   R,1,1e-4    ! Площадь поперечного сечения стержней AREA=1e-4 (произвольная) 
10  MP,EX,1,2E11 ! Модуль Юнга материалов стержней EX=2e11 (произвольный) 
11  S3=SQRT(3) 
12  N,1,0,0       ! Определяем узлы по координатам на плоскости 
13  N,2,2,0 
14  N,3,4,0 
15  N,4,2.5,S3/2 
16  N,5,1,S3 
17  E,1,2         ! Определяем элементы текущего типа TYPE=1 (LINK1) 
18  E,2,3         !  с текущими наборами констант REAL=1 
19  E,3,4         !  и материальных свойств MAT=1 
20  E,4,5         !  по номерам граничных узлов 
21  E,5,1 
22  E,5,2 
23  E,2,4 
24  FINISH 
25  /SOLU 
26  ANTYPE,STATIC     ! Статический анализ 
27  !  Определяем шарнирные опоры в узлах 
28  D,1,UY,0,,3,2     ! UY=0 в узлах 1 и 3  
29  D,3,UX            ! UX=0 в узле 3 
30  !  Задаем силы в узлах 
31  F,1,FY,-1E3,,3,2  ! FY=-1e3 в узлах 1 и 3 
32  F,4,FY,-2E3,,5    ! FY=-2e3 в узлах 4 и 5 
33  SOLVE             ! решаем СЛАУ 
34  FINISH 
35  /POST1 
36  /OUTPUT,F_R,res   ! Направляем вывод в файл <F_R>.res 
37  PRRSOL            ! Печатаем опорные реакции 

38  PRESOL,SMISC,1    ! Печатаем усилия в стержнях 
39  /OUTPUT 
40  FINISH 
41  /EXIT           
 
 В результате работы программы в файле SMs1.res будут находиться 
искомые величины. Сравнение полученных результатов с ответами к зада-
че № 4.67 (5.7) из [1], как показывают таблицы 1.1 и 1.2, позволяют сделать 
выводы о правильности работы программы и прекрасной точности расче-
тов ферм на ANSYS. 
 

Таблица 1.1. 
 

Номер узла 1 3 
Реакции в узлах (кН) Ответ из задачника 3.25 2.75 

 ANSYS 3.25 2.75 
 

Таблица 1.2. 
 

Номер стержня 1 2 3 4 5 6 7 
Усилия 

(кН) 
Ответ из 
задачника 

1.3 3.03 -3.5 -2.5 -2.6 1.73 -1.73 

 ANSYS 1.299 3.0311 -3.5 -2.5 -2.5981 1.7321 -1.7321 
 

 Дадим дополнительные комментарии по программе SMs1.inp, а так-
же по общей методике расчета ферм на ANSYS. 
 Операторами с номерами 6 и 11 вводятся и инициализируются ска-
лярные параметры F_R и S3. Параметру F_R присваивается строковая кон-
станта ‘SMs1’, а параметру S3 - действительная константа 3 . Техника ис-
пользования параметров подробно описана в руководстве [4]. 
 Оператор 8 определяет элементный тип TYPE со ссылочным номе-
ром 1 (TYPE=1) как стержневой КЭ LINK1; оператор 9 - набор элементных 
констант (Real Constants) со ссылочным номером 1 (REAL=1), состоящий 
здесь из одной константы 1e-4. Оператор 10 задает физическое свойство 
EX=2e11 и относит это значение к набору материальных свойств (Material 
Properties) со ссылочным номером 1 (MAT=1). 
 Операторы 12-16 последовательно определяют узлы (Nodes) с номе-
рами 1-5 по двум координатам X и Y в текущей глобальной декартовой 
системе координат. Операторы 17-23 задают конечные элементы по номе-
рам узлов. В отличие, например, от команды определения узлов N, в ко-
манде E нет поля для номера элемента. При выполнении команды E созда-
ваемому КЭ автоматически присваивается номер КЭ, равный максималь-
ному номеру из множества занумерованных элементов + 1. При создании 
элемента с ним связываются также текущие или принятые по умолчанию 
значения из множеств MAT, REAL, TYPE и ESYS. (Последнее множество 

есть множество ссылочных номеров элементных координатных систем.) 
Так как определены MAT=1, REAL=1, TYPE=1, и эти значения являются 
текущими по умолчанию, то все элементы, создаваемые по операторам 17-
23, будут иметь эти же свойства. Заметим, что материальные свойства MP, 
константы R и элементные типы ET определяются независимо различными 
командами, и правильность их задания для отдельных КЭ является задачей 
пользователя. 
 После создания конечно-элементной модели функции препроцессора 
PREP7 выполнены. В решателе SOLUTION определяется статический тип 
анализа (оператор 26) и задаются условия закрепления и силовые факторы. 
Согласно принятой в ANSYS концепции, все граничные условия и силовые 
факторы трактуются как «нагрузки» (Loads). Различаются следующие ти-
пы нагрузок: 1) DOF Constraint Loads; 2) Force Loads; 3) Surface Loads; 4) 
Body Loads и 5) Inertia Loads. Большинство команд для первых четырех 
классов начинаются с той же самой буквы, с которой начинается и назва-
ние соответствующего класса (D, F, S, B). Нагрузки также подразделяются 
на два следующих подкласса: нагрузки, приложенные к конечно-
элементной модели (Finite Element Loads), и нагрузки, приложенные к гео-
метрической модели (Solid Model Loads). Поскольку в данном примере в 
построении геометрической модели не было необходимости, то здесь ис-
пользуются только нагрузки, приложенные к конечно-элементной модели. 
Команды типа DOF Constraint Load Commands связывают определенные 
степени свободы (DOF - degree of freedom) с конкретными значениями. В 
механике это обычно есть главные граничные условия, накладываемые на 
перемещения, и возможно, на углы поворота. В нашем примере мы имеем 
два шарнира в узлах 1 и 3 (рис. 1.1), причем в узле 1 нет ограничений на 
перемещения по оси OX. Степенями свободы в каждом узле для КЭ LINK1 
являются компоненты вектора перемещений UX и UY. Тогда, очевидно, 
что для удовлетворения условиям закрепления в узле 1 надо положить 
UX=0; а в узле 3 - UX=UY=0. Эти цели и достигаются операторами 28 и 29 
программы SMs1.inp. Отметим, что вместо сложного формата оператора 28 
можно было бы использовать два оператора (через знак $ можно записы-
вать несколько команд в одной строке) 
 D,1,UY,0   $   D,3,UY,0 
причем, как и в операторе 29, в соответствии с правилами умолчания 
ANSYS нуль в поле значений VALUE компоненты DOF можно опустить: 
D,1,UY и т.д. 
 Аналогичным образом, операторы 31 и 32 задают значения сосредо-
точенных сил (Force) в определенных узлах. Так как сила является векто-
ром, то для плоской задачи задаются ее компоненты по осям X и Y. В на-
шем примере внешние силы направлены против оси OY, и следовательно, 
имеют ненулевыми лишь компоненты FY, причем со знаком «минус». Как 

и для команд D, операторы 31 и 32 можно было бы использовать в более 
простой форме: 
 F,1,FY,-1e3  $  F,3,FY,-1e3 
 F,4,FY,-2e3  $  F,5,FY,-2e3 
 Команда SOLVE формирует и решает систему линейных алгебраиче-
ских уравнений (СЛАУ) МКЭ, соответствующую созданной конечно-
элементной модели и типу анализа, а также записывает получаемые ре-
зультаты в базу данных *.db и файл вывода. Команды опций вывода 
(OUTRES и OUTPR) позволяют определять различные типы данных, за-
писываемых в базу или выводимых на печать. По умолчанию принимается 
OUTRES,ALL, т.е. сохранение всех данных (одношагового) статического 
анализа в *.db, и подавление вывода на печать. 
 Поскольку все данные в результате решения оказываются доступны-
ми для обработки, используя главный постпроцессор POST1 и операторы 
37, 38, мы можем вывести опорные реакции и усилия в стержнях. Отметим, 
что целью решения задач является именно получение определенных ре-
зультатов. О том, какие результаты и в каком виде надо получить, следует 
думать в самом начале решения задачи. В ряде случаев для этого необхо-
димо устанавливать специальные опции вывода, определять переменные 
для графиков, проводить дополнительные сервисные вычисления и т.п. 
Для нашего примера достаточны текстовые результаты. При этом реакции 
можно вывести по команде 37, а усилия в стержнях - по команде 38. Отно-
сительно команды 38 отметим, что поля SMISC,1 определяют именно уси-
лия в стержнях для КЭ LINK1 согласно документации по КЭ LINK1 [6]. 
Наконец, команда 36 позволяет назначить вывод в желаемый файл, а ко-
манда 39 - вернуть вывод в стандартный файл вывода *.out. Специальный 
файл вывода SMs1.res используется в данной программе для отсечения 
многих ненужных сообщений, создаваемых в процессе выполнения про-
граммы. 
 При желании графическое изображение деформированного состоя-
ния фермы можно получить, используя в препроцессоре команду: 
 PLDISP,1   !  Показать деформир. состояние вместе с недеформированным 
Однако, перед этим желательно задать вывод графики в файл графического 
ANSYS-формата, например, в файл <f_r>.grph по команде: 
 /SHOW,f_r,grph 
 Изменяя соответствующим образом файл SMs1.inp, можно создавать 
программы для расчета разнообразных ферм. При этом для ANSYS безраз-
лично, рассматривается ли статически определимая ферма, или статически 
неопределимая. Естественно, что в последнем случае существенны значе-
ния констант REAL и материальных свойств MAT. 
 Если в ферме имеются стержни с различными свойствами, то воз-
можно применение следующей стратегии. Пусть, например, стержень 2 

выполнен из алюминия с модулем Юнга 110 710EX= ⋅. )  (н/м2  и площадью 
поперечного сечения −= ⋅ 4210AREA  (м). Тогда при задании наборов REAL 
и MAT следует также определить новые REAL и MAT с номерами, равны-
ми, например, 2: 
 R,2,2e-4 
 MP,EX,2,0.7e11 
 При определении элементов необходимо активизировать соответст-
вующие наборы REAL и MAT. Таким образом, команды задания элементов 
17-19 нужно видоизменить, например, так: 
 E,1,2 
 REAL,2  $  MAT,2  $  E,2,3 
 REAL,1  $  MAT,1  $  E,3,4 
 (Для элемента E,1,2 текущими значениями атрибутов по умолчанию 
являются значения, равные 1: MAT=1, REAL=1, TYPE=1, ESYS=1. Перед 
созданием второго элемента (E,2,3) следует изменить значения MAT и 
REAL на 2, а для других элементов (E,3,4 и т.д.) - снова вернуться к значе-
ниям MAT и REAL, равным 1.) 
 Если требуется определить узловые силы в узлах фермы, то в коман-
дах вывода (например, после команды 37), можно добавить команду, пред-
назначенную для вывода узловых нагрузок (для данного примера - сил): 
 PRNLD 
 Для вывода осевых напряжений, существенных для задач сопротив-
ления материалов, можно  в решателе SOLUTION использовать команду: 
 OUTPR,BASIC,1 
но тогда печать осевых напряжений SAXL будет осуществлена в стандарт-
ный файл вывода *.out. Более сложный, но и более изящный способ - ис-
пользование следующего фрагмента внутри постпроцессора: 

/NOPR      ! подавление ненужного вывода 
N_EL=7               ! параметр для количества элементов 
*DO,I,1,N_EL     ! цикл по элементам 
  ESEL,S,ELEM,,I,I  ! выбор элемента с номером I 
  ETABLE,SIGAX,LS,1 ! из таблицы ETAB выбираются указатели 
                                       ! SIGAX на SAXL согласно документации 
                                       ! КЭ LINK1 
  *GET,SS,ELEM,I,ETAB,SIGAX ! Определяем SS по указателю 
  /GOPR      ! активизация вывода 
  *VWRITE,I,SS    ! выводим номер элемента I и напряжение SS 
  (1x,' I=',F4.0,' SIGMA_AX=',e10.4) 
  /NOPR 
*ENDDO             ! конец цикла по элементам 
При этом выводная информация будет представлена в наиболее ко-

роткой и удобной форме. (Добавьте в этот фрагмент также команды для 
вывода усилий в стержнях!) 

Другим способом вывода усилий и напряжений в стержнях являются 
следующие команды: 



 11

ETABLE,FORAX,SMISC,1  ! FORAX - указатель на SMISC,1 
ETABLE,SIGAX,LS,1           !  SIGAX - указатель на LS,1 
PRETAB,FORAX,SIGAX     !  Печать FORAX и SIGAX 

 Наконец, расчет пространственных ферм отличается от расчета пло-
ских ферм только использованием стержневых КЭ LINK8 вместо LINK1 
(значения LS,1 и SMISC,1 и для КЭ LINK8 определяют соответственно 
осевые напряжения и усилия). Естественно, что узлы теперь требуется за-
давать по трем декартовым координатам. Например, узел 1 с координатами 
(2, 0, 1.5) задается по команде: 
 N,1,2,0,1.5    или     N,1,2,,1.5 
 В заключение разбора примера расчета фермы подчеркнем, что фи-
зические свойства, константы из множеств R и геометрия должны быть за-
даны в единой физической системе единиц. Итоговые результаты будут 
тогда представлены в той же системе единиц. (Здесь была использована 
система СИ.) 
 Ниже приводятся варианты заданий для самостоятельного выполне-
ния. 
 Варианты задания № 1. Рассчитать ферму на ANSYS, используя 
данные, приведенные в указанных задачах. Определить опорные реакции и 
усилия в стержнях. Для задач из [1] и [3] сравнить полученные результаты 
с приведенными в ответах к соответствующим задачам. (При выполнении 
заданий из [3] считать известными из ответов задач силы или площади се-
чений и определить также осевые напряжения.) 

Таблица 1.3. 
Вариант 1а. 

 
№  задания 1a.1 1a.2 1a.3 1a.4 1a.5 1a.6 1a.7 1a.8 1a.9 
№ задачи 4.45 

[1] 
4.46 
[1] 

4.65 
[1] 

4.66 
[1] 

4.68 
[1] 

4.69 
[1] 

4.70 
[1] 

4.71 
[1] 

4.72 
[1] 

 
1a.10 1a.11 1a.12 1a.13 1a.14 1a.15 1a.16 1a.17 1a.18 1a.19 1a.20 
4.73 
[1] 

1.32 
[3] 

1.47 
[3] 

1.51 
[3] 

1.53 
[3] 

1.54 
[3] 

1.71 
[3] 

1.73 
[3] 

1.75 
[3] 

1.77 
[3] 

1.78 
[3] 

 
Варианты 1б - 1г: 

 
1б)   1б.1 - 1б.30  -  С1.1 - С1.30 [2]; 
1в)   1в.1 - 1в.30  -  С3.1 - С3.30 [2]; 
1г)    1г.1 - 1г.30   -  С8.1 - С8.30 [2]. 

2. Плоский изгиб балок 
 

Рассмотрим классическую задачу сопротивления материалов о пря-
мом изгибе в одной плоскости упругой многопролетной балки с кусочно-
постоянными изгибными жесткостями. Внешними силовыми факторами: 
вызывающими такой изгиб, могут являться сосредоточенные силы, момен-
ты и распределенные нагрузки с кусочно-линейным изменением интенсив-
ности. 

Требуется составить программу расчета конкретной задачи из данно-
го класса так, чтобы при небольших изменениях можно было бы рассчиты-
вать и другие подобные задачи. Сервис программы должен быть достаточ-
но независимым от частной задачи и обеспечивать нахождение реакций 
опор,  значений перерезывающих сил, изгибных моментов, изгибных на-
пряжений и формы изогнутой оси балки. Результаты следует представить в 
текстовом и графическом видах. 

В качестве примера возьмем балку из задачи 9.1 сборника задач по 
сопротивлению материалов [3], показанную на рис.2.1. Данная балка отне-
сена к декартовой системе координат Oxyz так, что ось x направлена вдоль 
оси балки, проходящей через центры тяжести поперечных сечений. Оси y и 
z направлены вдоль главных осей инерции сечения, а начало отчета распо-
ложено в левом конце балки. Прямой поперечный изгиб здесь осуществля-
ется в плоскости Oxy. В соответствии с данными, взятыми из условия и 
решения задачи 9.1, примем: 

 
   y 
 
  q 

     M 
 
   O 
            x 
 

z        a    2a 
 
     Рис.2.1. 
 

( )a м м см сталь= = = ⋅ = ⋅2 2 2 2 106 2 м  q  T  M  T  E  кГ  ; / ; ; / ; A = 28 9 2. ; см

Jzz = ⋅2030 4 см ; h = 20 см.  Значение осевого момента инерции J zz , площади 
поперечного сечения A  и высоты h  взяты из характеристик двутавра 
стальной балки с номером профиля 20а. 

 Вариант программы для пакетного режима ANSYS, предназначенной 
для решения данной задачи, показан ниже. Отметим, что подходящим КЭ 
для задач плоского изгиба является балочный элемент BEAM3. Для этого 
элемента силы F считаются положительными, если их направления совпа-
дают с положительным направлением оси Oy. Наоборот, распределенные 
нагрузки q считаются положительными, если их направления противопо-
ложны направлению +y. Наконец, моменты zM M=  положительны при на-
правлениях вращения против часовой стрелки. Степенями свободы КЭ 
BEAM3 являются узловые перемещения UX, UY и угол поворота ROTZ. 
 Балку необходимо разбивать на КЭ BEAM3 так, чтобы в пределах 
каждого элемента были бы постоянными изгибные жесткости zzEJ  , рас-
пределенные нагрузки менялись бы максимум линейно, а внутри элемен-
тов не находились бы опоры и точки приложения активных сил и момен-
тов. Для рассматриваемой задачи достаточно всего два элемента BEAM3. 
Здесь, однако, использованы три элемента одинаковой длины. Добавление 
«лишнего» элемента позволяет улучшить графические представления ре-
зультатов, поскольку при построении графиков внутри отдельного элемен-
та используется только 9 промежуточных точек (KEYOPT(9)=9 для 
BEAM3). 
 
 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                 Файл SMs2.inp 
/com,    ПРОГРАММА РАСЧЕТА ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,           ПО ТЕМЕ: "ПЛОСКИЙ ИЗГИБ БАЛКИ" 
/com,                                                (версия 1.4) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 9.1.    С.228-229.) 
/com, 
/com,   Текстовые результаты сохраняются в файле SMs2.res 
/com,   Графические результаты выводятся в файл  SMs2.grph 
/com, 
/prep7                     ! Входим в препроцессор /PREP7 
f_r=’SMs2’ 
/show,f_r,grph       ! В режиме /BATCH задаем вывод графики 
                               ! в файл <f_r>.grph 
/title, BENDING OF THE BEAM 
                               ! Все данные в системе СГС 
et,1,beam3       ! Элементы с TYPE=1 - , балочные КЭ BEAM3 
keyopt,1,9,9     ! KEYOPT(9)=9 для BEAM3 - 
                         !   вывод в 9 промежуточных точках элемента 
r,1,28.9,2030,20 ! REAL CONST. типа REAL=1 : (двутавр N 20а) 
                           !   площадь поперечного сечения AREA=28.9 , 

                           !   момент инерции IZZ=2030 , 
                           !   толщина HEIGHT=20 
mp,ex,1,2e6       ! MAT. PROP. типа MAT=1: модуль Юнга EX=2e6 
n_el=3                ! Вводим параметр для количества элементов 
abeam=200        ! abeam - параметр длины первого пролета балки 
n,1                     ! Определяем узлы по координатам 
n,2,abeam 
n,3,2*abeam 
n,4,3*abeam 
e,1,2                  ! Определяем элементы BEAM3 с узлами 1 и 2, 
e,2,3                  !                                                                        2 и 3, 
e,3,4                  !                                                                        3 и 4 
save 
finish   
/solu                 ! Входим в решатель /SOLUTION 
antype,static    ! Статический тип анализа 
d,1,uy               ! Задаем условия закрепления: 
d,4,ux               !  UY=0 в узле 1 (шарнир на "колесиках"), 
d,4,uy               !  UX=UY=ROTZ=0 в узле 4 (жесткая заделка) 
d,4,rotz            ! 
sfbeam,1,,pres,20 ! Задаем постоянную распределенную нагрузку 
                             ! на элементе 1  PRES=20 
                             ! (положительным значением PRES считается 
                             !  направление против оси +Y) 
f,2,mz,-2e5           ! Задаем момент MZ=-2e5 в узле 2 
                             ! (положительным считается MZ, 
                             !  направленный против часовой стрелки) 
     ! Задание сил FY или FX проводится аналогично с заменой 
     ! MZ на FY или FX (сила FY положительна, если направлена 
     ! по оси +Y, т.е. наоборот, по сравнению с PRES) 
solve             ! Решаем СЛАУ МКЭ 
finish   
/post1            ! Входим в постпроцессор /POST1 
set,1,1           ! Считываем данные о решении 
   ! Далее следует вывод перерезывающих сил, изгибающих моментов, 
   ! уравнения изогнутой оси балки и изгибных напряжений 
   ! в 11 равноотстоящих точках каждого элемента 
*dim,x_coord,array,n_el*11  ! массив для X-координат 
*dim,q_y,array,n_el*11         ! массив для перерезывающих сил 
*dim,m_z,array,n_el*11        ! массив для изгибающих моментов 
*dim,u_y,array,n_el*11         ! массив для прогибов балки 
*dim,sigma,array,n_el*11     ! массив для изгибных напряжений 
/output,f_r,res                       ! Задаем текстовый вывод 
                                               ! в файл <f_r>.res 
/com, 
/com, *********************************************************** 
prrsol                      ! Вывод реакций опор 
/com, 
/com, *********************************************************** 

/com,     X     !    Q_Y   !     M_Z   !     U_Y   !   Sigma_YY 
/com, *********************************************************** 
/nopr 
*do,i,1,n_el         ! цикл по элементам 
 n1_nood=nelem(i,1)        ! Определяем номера граничных узлов 
 n2_nood=nelem(i,2)        ! n1_nood и n2_nood i-ого элемента 
 nx1_nood=nx(n1_nood)      ! Определяем X-координаты граничных 
 nx2_nood=nx(n2_nood)      ! узлов nx1_nood и nx2_nood 
 l_elem=nx2_nood-nx1_nood  ! Находим длину i-ого элемента 
 hx=l_elem/10               ! hx - расстояние между 11 точками 
                                     !      вывода в i-ом элементе 
 uy1=uy(n1_nood)           ! uy1, uy2 - перемещения в узлах 
 uy2=uy(n2_nood)           !                  i-ого элемента 
 tz1=rotz(n1_nood)         ! tz1, tz2 - углы поворота в узлах 
 tz2=rotz(n2_nood)         !                i-ого элемента 
 esel,s,elem,,i,i          ! Выбираем элемент с номером i 
 /gopr 
 *vwrite,i                 ! Выводим текущий номер элемента 
 (1x,'     element N = ',f10.0) 
 /nopr 
 *do,ii,1,11       ! цикл по точкам элементов 
  j=ii+(i-1)*11            ! j - текущий номер элементов массивов 
  xx=nx1_nood+hx*(ii-1)    ! xx - X-координата j-ой точки 
  x_coord(j)=xx 
  ksi=(ii-1)*0.2-1         ! ksi - значение, соответствующее xx 
                                    !       в локальной системе координат 
     ! Вычисляем u_y(j) в j-ой точке c использованием 
     ! функции аппроксимации перемещений на элементе 
  uu=0.5*(uy1+uy2+(uy2-uy1)*ksi*(3-ksi*ksi)*0.5) 
  uu=uu+0.125*l_elem*(tz1*(1-ksi)-tz2*(1+ksi))*(1-ksi*ksi) 
  u_y(j)=uu 
  etable,sig,ls,ii*3-1             ! Из таблицы ETABLE выбираем 
  etable,qq_y,smisc,ii*6-4   ! указатели на SBYT, MFORY, MMOMZ 
  etable,mm_z,smisc,ii*6     ! для sigma, q_y, m_z 
                                            ! согласно документации по BEAM3 
                                            ! при KEYOPT(9)=9 
  *get,ss,elem,i,etab,sig        ! Определяем sigma, q_y, m_z 
  *get,qq,elem,i,etab,qq_y   ! по найденным указателям 
  *get,mm,elem,i,etab,mm_z 
  qq=-qq                     ! Перерезывающая сила в ANSYS имеет 
                                   ! противоположный знак по сравнению 
                                   ! с принятым для эпюр перерезывающих 
                                   ! сил 
  sigma(j)=ss 
  q_y(j)=qq 
  m_z(j)=mm 
  /gopr 
  *vwrite,xx,qq,mm,uu,ss     ! Вывод найденных значений 
  (1x,e10.4,' ',e10.4,'  ',e10.4,'  ',e10.4,' ',e10.4) 

  /nopr 
 *enddo 
*enddo 
/output 
esel,all 
finish 
/post26                          ! Входим в построцессор /POST26 
store,alloc,n_el*11        ! Подготавливаем место для размещения 
                                      ! переменных для графиков 
/xrang,nx(nelem(1,1)),nx(nelem(n_el,2)) ! изменение переменной X 
                                                                  !  совпадает с длиной балки 
/axlab,x, X-coordinate     ! Задаем подпись для оси абсцисс 
/color,curve,whit,1,6      ! Цвет кривых на графиках - белый 
/color,grid,whit              ! Цвет сетки - белый 
/plopts,info,off               ! Íå âûâîäèòü ñëåâà äîïîëíèò. èíôîðìàöèþ 
/plopts,frame,off            ! Íå ïîêàçûâàòü îêîííóþ ðàìêó 
grid,1                              ! Задаем вывод сетки на графиках 
vput,x_coord(1),1          ! Помещаем в переменные 1-5 
vput,q_y(1),2                 ! соответствующие массивы 
vput,m_z(1),3 
vput,u_y(1),4 
vput,sigma(1),5 
xvar,1                     ! Задаем переменную по оси X 
/output,f_r,res,,append        ! Восстанавливаем тестовый вывод 
                                              ! в файл <f_r>.res с записью 
                                              ! в конец файла 
/com, 
/com, *********************************************************** 
/com, 
/com,  VARI 2  -  Q_y 
/com,  VARI 3  -  M_z 
/com,  VARI 4  -  U_y 
/com,  VARI 5  -  Sigma_YY 
/com,  TIME    -  X-coordinate 
/com, 
/com, *********************************************************** 
/com, 
extrem,2,5    ! Вывод экстремальных значений переменных 2-5 
/output 
  ! Далее последовательно задаем надписи по осям ординат 
  ! и записываем в графический файл <f_r>.grph графики 
  ! перерезывающих сил Q_Y, изгибающих моментов M_Z, 
  ! перемещений U_Y, изгибных напряжений Sigma_YY 
/window,1,ltop       ! Îêíî 1 - ñëåâà ââåðõó 
/axlab,y,Shear force Q_y 
plvar,2 
/window,1,off        ! Ñäåëàòü îêíî 1 ïàññèâíûì 
/noerase                ! Íå î÷èùàòü ñîäåðæèìîå îêîí 
/window,2,rtop       ! Îêíî 2 - ñïðàâà ââåðõó 

/axlab,y,Bending moment M_z 
plvar,3 
/window,2,off 
/window,3,lbot       ! Îêíî 3 - ñëåâà âíèçó 
/axlab,y,Displacement U_y 
plvar,4 
/window,3,off 
/window,4,rbot       ! Îêíî 4 - ñïðàâà âíèçó 
/axlab,y,Bending stress at the +Y side  
plvar,5 
finish 
/exit     ! Окончание работы программы 
 
 Приведенная программа снабжена подробными комментариями, что 
позволяет разобраться в принятой при ее составлении стратегии. Результа-
том выполнения программы являются текстовый файл SMs2.res и графиче-
ский файл SMs2.grph, записанный в специальном формате ANSYS. Файл 
SMs2.grph содержит четыре следующих графика (Рис.2.2): эпюр перерезы-
вающих сил (Shear force Q_y), моментов (Bending moment M_z), уравнения 
изогнутой оси (Displacement U_y) и изгибных напряжений (Bending stress 
at the +Y side). 
 

 
Рис. 2.2. 

 В текстовом файле SMs2.res содержатся данные об опорных реакци-
ях и численных значениях для всех четырех эпюр вместе с их максимума-
ми и минимумами. В приводимых ниже таблицах 2.1 - 2.4 дается сравнение 
основных величин, рассчитанных на ANSYS по МКЭ и приведенных в [3] 
в результате аналитического решения задачи. Как видно из этих таблиц, 
полученные результаты прекрасно согласуются с аналитическими. 
 

Таблица 2.1. 
Значения опорных реакций 

 
 R_y при x=0 (кГ) R_y при x=3a (кГ) M при x=3a (кГсм) 
[3] 2570 1430 -2.56е+5 
ANSYS 2574.1 1425.9 -2.5556е+5 
 

Таблица 2.2. 
Значения поперечных сил Q_y (кГ) в характерных точках 

 
 x=0 x=a x=3a 
[3] 2570 -1430 -1430 
ANSYS 2574 -1426 -1426 
 

Таблица 2.3. 
Значения изгибающих моментов M_z (кГсм) в характерных точках 

 
x (см) x=128.5 x=200 (слева) x=200 (справа) x=600 

[3] 1.65e+5  3.14e+5 -2.56e+5 
ANSYS 1.649e+5 1.148e+5 3.148e+5 -2.556e+5 
 

Таблица 2.4. 
Значения максимальных прогибов U_y (см) 

 
 x (см) U_y (см) 
[3] 242 -1.35 
ANSYS 240 -1.348 
 
 
 Используя программу SMs2.inp, легко получить программы для ре-
шения других аналогичных задач. Изменения будут затрагивать только 
операторы препроцессора и решателя, которых имеется совсем не большое 
число. 
 (Техника задания промежуточных шарниров для составных балок 
описана в следующем разделе.) 

 Варианты задания № 2. Для конкретной балки, приведенной в со-
ответствующей задаче из [3], рассчитать на ANSYS реакции опор, 
перерезывающие силы, изгибающие моменты, изгибные напряжения и 
уравнение изогнутой оси балки. Определить максимальные и 
минимальные значения всех определяемых величин. Сравнить полученные 
результаты с приведенными в ответах к соответствующих задачах из [3]. 
Результаты представить в текстовых и графических формах. Номера задач 
из [3] приводятся ниже. 
 

Таблица 2.5. 
 
№  задания 2.1 2.2 2.3 2.4 2.5 2.6 2.7 2.8 2.9 
№ задачи из 
[3] 

9.2 9.3 9.4 9.5 9.6 9.32а 9.9а 9.9д 9.9л 

 
2.10 2.11 2.12 2.13 2.14 2.15 2.16 2.17 2.18 2.19 2.20 
9.9м 9.10г 9.10д 9.30а 9.30б 9.22а 9.22в 9.22г 9.22д 9.9м 9.10г 
 
2.21 2.22 2.23 2.24 2.25 2.26 2.27 2.28 2.29 2.30 
9.33а 9.33б 9.33в 9.33г 9.33д 9.33е 9.30в 9.30г 9.30д 9.30е 
 
 Определить также существенность использования модели балки Ти-
мошенко для данных задач, сравнив расчеты по двум моделям балки 
(обычной модели и модели балки Тимошенко). Для определения балки 
Тимошенко нужно дополнительно задать коэффициент сдвига sf  как зна-
чение поля SHEARZ в команде R следующего формата: R, NREAL, AREA, 
IZZ, HEIGHT, SHEARZ ; а также модуль сдвига GXY по команде MP. Зна-
чение коэффициента сдвига sf  можно приближенно принять равным: 6/5 - 
для прямоугольника; 10/9 или 32/27 - для круга; 2 - для тонкостенного 
кольца; 12/5 - для тонкостенного полого квадрата; A/(hd) - для двутаврово-
го профиля, где A - площадь, h - высота, d - толщина поперечного сечения 
двутавра. 

3. Расчет рам 
 

Расчет рам на ANSYS на этапах программирования в препроцессоре 
и решателе во многом аналогичен решению задач об изгибе балок. Для 
плоских рам по-прежнему достаточен КЭ BEAM3, так как он может рабо-
тать как на изгиб, так и на растяжение-сжатие. В пространственном случае 
необходимо использовать КЭ BEAM4. 

Отдельное внимание уделим случаю наличия промежуточных шар-
ниров. Этот случай важен и для задач об изгибе балок, но в разделе 3 не 
рассматривался. Пусть два конечных элемента BEAM3 имеют общую кон-
цевую точку, в которой располагается шарнир. Очевидно, что тогда в этой 
точке оба конечных элемента должны иметь одинаковые перемещения UX 
и UY, но углы поворота ROTZ могут быть, вообще говоря, различными. 
Именно поэтому в общей концевой точке для этих двух конечных элемен-
тов не может быть общего узла! Здесь должны существовать два различ-
ных узла с одинаковыми геометрическими координатами. Кроме того, для 
таких узлов нужно связать между собой по отдельности узловые степени 
свободы UX и UY так, чтобы обеспечить их равенство. Данная процедура 
может быть осуществлена в ANSYS в препроцессоре по командам CP сле-
дующим образом: 

CP, NSET1, UX, NODE1, NODE2 
CP, NSET2, UY, NODE1, NODE2 

где NSET1 и NSET2 - ссылочные имена множеств связанных степеней сво-
боды UX и UY соответственно, а NODE1 и NODE2 - номера узлов с одина-
ковыми координатами, в которых располагается промежуточный шарнир. 
Подчеркнем, что ссылочные номера NSETJ должны быть различными для 
всех различных множеств связанных степеней свободы. 
 Описанная схема может быть осуществлена и в программе beam1.inp 
из раздела 2 для задач об изгибе балок с промежуточными шарнирами. Не-
сложно увидеть, что при этом все операторы из постпроцессоров /POST1 и 
/POST26 в программе beam1.inp можно оставить без изменения. 
 Далее приводится пример программы из файла SMs3.inp, предназна-
ченной для расчета плоской рамы из задачи 9.1 [3] (Рис. 3.1 и 3.2). 
 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                 Файл SMs3.inp 
/com,    ПРОГРАММА РЕШЕНИЯ ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,           ПО ТЕМЕ: "РАСЧЕТ ПЛОСКИХ РАМ" 
/com,                  (версия 1.1) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 9.36.    С.246-247.) 

/com, 
/com,   Текстовые результаты сохраняются в файле SMs3.res 
/com,   Графические результаты выводятся в файл SMs3.grph 
/com, 
/title, PLANE FRAME 
F_R='SMs3' 
/prep7           ! Входим в препроцессор /PREP7 
/show,F_R,grph   ! В режиме /BATCH задаем вывод графики 
                 ! в файл <F_R>.grph 
                 ! Все данные в системе СГС 
et,1,beam3       ! Элементы с TYPE=1 - , балочные КЭ BEAM3 
r,1,37.5,3800,24 ! REAL CONST. типа REAL=1 : (двутавр N 20а) 
                 !   площадь поперечного сечения AREA=28.9 , 
                 !   момент инерции IZZ=2030 , 
                 !   толщина HEIGHT=20 
mp,ex,1,2e6      ! MAT. PROP. типа MAT=1: модуль Юнга EX=2e6 
lr=500   ! lr, ar, cr - геометрические параметры 
ar=400 
cr=300 
n_fill=16      ! число КЭ, генерируемых  
               ! на участке с распределенной нагрузкой 
n,1              ! Определяем узлы по координатам 
n,2,,lr 
n,3,,lr 
n,3+n_fill,ar,lr 
n,4+n_fill,ar,lr-cr 
fill,3,3+n_fill,n_fill-1  ! Генерируем n_fill-1 равноотстоящих узлов 
                          ! между узлами 3 и 3+n_fill 
e,1,2            ! Определяем элементы BEAM3 
*do,i,3,3+n_fill 
  e,i,i+1 
*enddo 
cp,1,ux,2,3 $ cp,2,uy,2,3   ! Связываем степени свободы UX и UY 
                  ! в совпадающих узлах 2 и 3 для задания 
                  ! промежуточного шарнира 
finish   
/solu            ! Входим в решатель /SOLUTION 
antype,static    ! Статический тип анализа 
d,1,uy           ! Задаем условия закрепления 
d,1,ux 
d,1,rotz 
d,4+n_fill,uy 
d,4+n_fill,ux            
*do,i,2,1+n_fill 
 sfbeam,i,,pres,40  ! Задаем распределенную нагрузку 
*enddo 
     ! Задание сил FY, FX и моментов MZ проводится аналогично 
     ! описанному в программе SMs2.inp 
solve             ! Решаем СЛАУ МКЭ 

save 
finish  
/post1 
 ! Согласно документации по КЭ BEAM3 при KEYOPT(9)=0 
 ! определяем указатели на значения нормальных и поперечных сил и 
 ! изгибающих моментов на концах элементов 
etable,mforxi,smisc,1  ! Smisc,1 - нормальная сила N в узле I 
etable,mforxj,smisc,7  ! Smisc,7 - нормальная сила N в узле J 
etable,mforyi,smisc,2  ! Smisc,2 - поперечная сила Q в узле I 
etable,mforyj,smisc,8  ! Smisc,8 - поперечная сила Q в узле J 
etable,mmomzi,smisc,6  ! Smisc,6 - изгибающий момент M в узле I 
etable,mmomzj,smisc,12 ! Smisc,12 - изгибающий момент M в узле J 
/output,F_R,res   ! Направляем вывод в файл <F_R>.res 
prrsol            ! Печатаем опорные реакции 
pretab,mforyi,mforyj       ! Печатаем поперечные силы 
pretab,mforxi,mforxj       ! Печатаем нормальные силы 
pretab,mmomzi,mmomzj       ! Печатаем изгибающие моменты 
/output 
/pnum,node,1  ! Нумеровать узлы при графическом показе 
eplot         ! Нарисовать элементы 
/pbc,u,,1     ! Показывать условия закрепления  
              ! для механических степеней свободы 
/pbc,cp,,1    ! Показывать связанные степени свободы 
  ! Для показа сил и моментов используются команды: 
/pbc,f,,1 
/pbc,m,,1 
/psf,pres,norm,1 ! Показывать распределенные нормальные нагрузки 
/pnum,elem,1  ! Нумеровать элементы 
/pnum,node,0  ! Не нумеровать узлы 
/number,1  ! Занумерованные объекты показывать  
           ! только различным цветом без нумерации 
/zoom,1,scrn,,,1.7 ! Масштабировать окно для удобства 
/pbc,all,,0      ! Не показывать механические условия закрепления, 
                 ! силы, моменты и связанные степени свободы 
/psf,pres,norm,0 ! Не показывать распределенные нагрузки 
/pnum,elem,0     ! Не нумеровать элементы 
/triad,off       ! Не показывать начало и систему координат  
plls,mforyi,mforyj,-2  ! Графический показ эпюры поперечных сил Q  
  ! Поперечные силы в ANSYS имеют противоположный знак  
  ! по сравнению с принятым в указанном выше задачнике 
plls,mforxi,mforxj,2   ! Графический показ эпюры нормальных сил N 
plls,mmomzi,mmomzj,2   ! Графический показ эпюры изгибающих моментов M 
finish  
/exit 
 
 В результате выполнения программы создается текстовый файл 
SMs3.res и графический файл SMs3.grph. Файл SMs3.grph содержит пять 
следующих цветных картинок: исходная рама с пронумерованными узлами 

(картинка 1 - Рис. 3.1); отмасштабированная рама, для которой различными 
цветами показаны конечные элементы, граничные условия, связанные сте-
пени свободы и нагрузки (2 - Рис. 3.2); эпюры поперечных (3 - Рис. 3.3) и 
нормальных сил (4 - Рис. 3.4) и изгибающих моментов (5 - Рис. 3.5). 
 

 
Рис. 3.1. 

 
 В таблице 3.1 дается сравнения основных результатов работы про-
граммы (файл SMs3.res) с ответами, приведенными в [3] при решении дан-
ной задачи. 
 

Таблица 3.1. 
 
 Y_A (кГ) X_A (кГ) M_A 

(кГсм) 
Y_B (кГ) X_B (кГ) M_{max} 

(кГсм) 
[3] 7620 8380 -2.55e+5 510 -510 7.26e+5 
ANSYS 7616.1 8383.9 -2.5594e+5 511.88 -511.88 7.2322e+5 
 
 Здесь Y_A, X_A - компоненты вектора реакции в левой опоре рамы 
(точка A), а Y_B, X_B - в правой (точка B). 

                  Рис. 3.2.                                         Рис. 3.3. 
 

 
                   Рис. 3.4.                                         Рис. 3.5. 
 
 Основное отличие программы SMs3.inp от программы SMs2.inp из 
раздела 2 состоит в иной постпроцессорной обработке результатов. Графи-
ческое построение эпюр здесь опирается на команду plls постпроцессора 
/POST1. Данная команда предназначена для графического показа заданных 
характеристик в виде эпюр. Однако, графики этих характеристик по ко-
манде plls максимум линейны в пределах одного КЭ и строятся по значе-
ниям этих характеристик на концах элементов. Между тем известно, что 
эпюры изгибающих моментов изменяются по квадратичному закону на тех 

участках, где заданы постоянные распределенные нагрузки. Поэтому, для 
более точной аппроксимации эпюр изгибающих моментов кусочно-
линейными функциями, стержни, нагруженные распределенными нагруз-
ками, необходимо дополнительно разбивать на ряд КЭ. В программе 
SMs3.inp участок с распределенной нагрузкой разбивается на n_fill КЭ. 
Для этого по команде fill генерируется дополнительно (n_fill-1) равноот-
стоящих узлов между граничными узлами участка с распределенной на-
грузкой. 
 Заметим, что этот же прием мог бы быть применен в программе 
SMs2.inp из раздела 2, так как балка является частным случаем рамы. На-
оборот, прием из программы SMs2.inp, состоящий в использовании опции 
KEYOPT(9)=9 для BEAM3, при построении эпюр рам по команде plls не 
дает желаемых результатов, так как для команды plls нет доступа к данным 
в девяти промежуточных точках, устанавливаемых по опции 
KEYOPT(9)=9. Отметим также, что многооконный вывод для программы 
SMs3.inp не представляется целесообразным, так как приводимая справа на 
экране важная здесь информация не включается в окна вывода, и следова-
тельно, может отражать результаты лишь для одного окна. 
 Вместо цветового показа значений в эпюрах можно было бы исполь-
зовать так называемый «векторный» показ, задав перед командами plls пе-
реход в «векторную» моду: 
 /number,0              ! Режим представления объектов в цвете и с номерами 
 /device,vector,on   !  Переход в «векторную» моду графического вывода 
(Проведите предлагаемый эксперимент и выберите для дальнейшего тот 
вид представления эпюр, который вам больше понравится!) 
 Варианты задания № 3. Для конкретной рамы, приведенной в соот-
ветствующей задаче из [3], рассчитать на ANSYS реакции опор, попереч-
ные и нормальные силы и изгибающие моменты. Определить максималь-
ные и минимальные значения всех определяемых величин. Результаты 
представить в текстовых и графических формах (эпюрах). Номера задач из 
[3] приводятся ниже в таблице 3.2. (Необходимые числовые данные при их 
отсутствии в условиях задать самостоятельно.) 
 

Таблица 3.2. 
 
№  задания 3.1 3.2 3.3 3.4 3.5 3.6 3.7 3.8 3.9 
№ задачи из [3] 9.40а 9.40б 9.43л 9.43ê 9.43и 9.43з 9.43ж 9.43е 9.43д 
 
3.10 3.11 3.12 3.13 3.14 3.15 3.16 3.17 3.18 3.19 3.20 
9.43г 9.43в 9.43б 9.43а 9.42б 9.42а 9.42в 9.41 9.39 9.38г 9.38в 
 
3.21 3.22 3.23 3.24 3.25 3.26 3.27 3.28 3.29 3.30 
9.38а 9.38б 9.37з 9.37ж 9.37е 9.37д 9.37г 9.37в 9.37б 9.37а 

4. Различные статические задачи изгиба  
и кручения стержневых систем 

 
 Возможности конечно-элементного моделирования задач техниче-
ской механики и сопротивления материалов с использованием ANSYS ко-
нечно же не ограничиваются темами, рассмотренными в предыдущих 
пунктах. В настоящем разделе обсуждаются некоторые другие типы задач 
и особенности их анализа в ANSYS. Изложение здесь более сжато, так как 
задания этого раздела предполагают самостоятельную разработку нужных 
программ, используя документацию по ANSYS [4-8] и приведенные при-
меры. 
 

4.1. Кручение стержней. При решении задач кручения сопротивле-
ния материалов на ANSYS подходящими КЭ являются КЭ BEAM4 и 
PIPE16. Эти элементы имеют степенями свободы узловые смещения UX, 
UY, UZ и углы поворота ROTX, ROTY, ROTX в двух концевых узлах. С их 
помощью можно решать разнообразные задачи растяжения-сжатия, изгиба 
и кручения стержневых систем (в т.ч. и для сложного напряженного со-
стояния). 

Элемент BEAM4 является обобщением плоского балочного элемента 
BEAM3 на трехмерный случай. Если ось x - продольная ось элемента, то 
моментом, вызывающим кручение, будет момент MX. Для задач кручения 
стержней требуется задавать модуль сдвига G=µ, а также специальную 
геометрическую характеристику поперечного сечения Jk  , которая для 
кругового сечения совпадает с полярным моментом инерции Jp  .В списке 
констант (Real Constants) для элемента BEAM4 [6] находим моменты 
инерции IZZ, IXX и IXX на второй, третьей и восьмой позициях списка со-
ответственно. При этом под моментом инерции IXX подразумевается кру-
тящий момент инерции. В [6] указано, что IXX либо задается явно, либо 
вычисляется по формуле: IXX=IYY+IZZ, если IXX не определен явно, или 
IXX=0. Для задания IXX можно использовать команду RMORE, R7,R8,..., 
R12 в виде RMORE,,IXX или команду RMODIF. 

Для стержней круглого сечения предпочтительнее использовать 
«трубчатый» КЭ PIPE16. Этот элемент представляет собой полую трубку, 
и для сплошного круглого стержня следует просто задать толщину трубки, 
равную радиусу сечения. Среди констант элемента (Real Constants) доста-
точно определить внешний диаметр OD и толщину стенки TKWALL : 
R,NSET,OD,TKWALL . 

В качестве примера рассмотрим задачу 4.33 из [3]. Здесь стальной 
стержень круглого сечения с защемленными торцами скручивается двумя 
равными и одинаково направленными моментами M=8e4 (кГ ⋅ см ). Геомет-

рические данные задачи следующие: диаметр стержня d=10 см; a=60 см; 
b=80 см. Требуется определить реакции в заделках и угол закручивания в 
сечении x=a+b/2 (рис. 4.1). 

 

 
Рис. 4.1. 

 

Программа SMs4.inp, решающая данную задачу, приводится ниже. 
 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                                                Файл SMs4.inp 
/com,    ПРОГРАММА РЕШЕНИЯ ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,                                        ПО ТЕМЕ: "КРУЧЕНИЕ" 
/com,                                                 (версия 1.1) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,      Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,                 (по задаче 4.33.    С.94.) 
/com, 
/com,       Два варианта конечных элементов - BEAM4 и PIPE16 
/com, 
/com,   Итоговые результаты записываются  
/com,   в стандартный файл вывода *.out 
/com, 
/title, TORSION OF A CIRCULAR ROD  
/prep7           ! Входим в препроцессор /PREP7 
                   ! Все данные в системе СГС 
pi=3.1416 
dr=10         ! dr - диаметр стержня 
ar=60         ! ar, b2r - геометрические параметры 
b2r=40 
jpol=pi*dr**4/32 ! константа для полярного момента инерции 
et,1,beam4       ! балочные КЭ BEAM4 
r,1,pi*dr**2/4,jpol/2   ! REAL CONST : 
                 !   площадь поперечного сечения AREA=pi*dr**2/4 , 
                 !   момент инерции IZZ=jpol/2  
rmore,,jpol      !  полярный момент инерции  
                        !  (8-ая REAL-константа для BEAM4) 
mp,ex,1,2e6      ! модуль Юнга EX=2e6 
mp,gxy,1,8e5     ! модуль сдвига GXY=8e5 
n,1              ! Определяем узлы по координатам 
n,2,ar 
n,3,ar+b2r 

n,4,ar+b2r*2 
n,5,(ar+b2r)*2 
*do,i,1,4        ! Определяем элементы BEAM4 
  e,i,i+1 
*enddo 
finish   
/solu            ! Входим в решатель /SOLUTION 
antype,static    ! Статический тип анализа 
d,1,all,0,,5,4    ! Задаем условия закрепления в узлах 
f,2,mx,8e4       ! Задаем крутящие моменты в узлах 
f,4,mx,8e4            
outpr,basic,1    ! Вывод основных результатов в стандартный файл 
solve                 ! Решаем СЛАУ МКЭ 
finish  
/prep7           ! Вариант с КЭ PIPE16 
et,1,pipe16      ! КЭ - трубки PIPE16 
r,1,dr,dr/2      ! REAL CONST : 
                 !   внешний диаметр сечения OD=dr, 
                 !   толщина стенок трубы TKWALL=dr/2  
                 !   (сплошная труба) 
finish 
/solu 
solve 
finish 
/exit 

 
В результате выполнения программы в файле jobname.out (jobname - 

имя создаваемых рабочих файлов) среди различной выводной информации 
найдем значения реактивных моментов MX=-8e4 (кГ ⋅ см ) и угол закручи-
вания ϕ=0.611155e-2 (рад)≈ ′21  , что совпадает с ответом к задаче из [3]. 

 
4.2. Кривые стержни. Для расчета кривых стержней круглого сече-

ния в ANSYS существует КЭ PIPE18. Этот элемент представляет собой ис-
кривленную полую трубку, отличаясь от КЭ PIPE16 именно наличием кри-
визны. При расчете кривых стержней с помощью элемента PIPE18 в списке 
констант КЭ требуется задать следующие величины: внешний диаметр OD, 
толщину стенки TKWALL и радиус кривизны RADCUR. Для этого исполь-
зуется команда: R,NSET,OD,TKWALL,RADCUR. При формировании эле-
мента PIPE18 необходимы три узла: два граничных узла и узел, опреде-
ляющий положение центра кривизны элемента. Если не требовать большо-
го сервиса, то программа расчета кривых стержней может быть похожа на 
приведенную ранее программу SMs4.inp. Например, для задачи № 11.18 из 
[3] такая программа может иметь следующий вид. 

 
/batch           ! Пакетный режим /BATCH 
/com, 

/com,                 Файл SMs5.inp 
/com,    ПРОГРАММА РЕШЕНИЯ ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,               ПО ТЕМЕ: "КРИВЫЕ СТЕРЖНИ" 
/com,                (версия 1.1 программы без построения эпюр) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 11.18.    С.331.) 
/com, 
/com,   Итоговые результаты записываются  
/com,   в стандартный файл вывода *.out 
/com, 
/title, CURVED BAR  
/prep7           ! Входим в препроцессор /PREP7 
                 ! Все данные в системе СГС 
dr=2          ! dr - диаметр стержня 
ar=10         ! ar - радиус кривизны 
et,1,pipe18      ! КЭ - искривленные трубки PIPE18 
keyopt,1,6,2     ! печать внутренних усилий 
r,1,dr,dr/2,ar   ! REAL CONST : 
                 !   внешний диаметр сечения OD=dr, 
                 !   толщина стенок трубы TKWALL=dr/2  
                 !   (сплошная труба) 
                 !   радиус кривизны RADCUR=ar 
mp,ex,1,2e6      ! модуль Юнга EX=2e6 
mp,gxy,1,8e5     ! модуль сдвига GXY=8e5 
n,1              ! Определяем узлы и элементы 
n,2,-ar 
n,3,0,ar 
n,4,ar 
e,2,3,1 $ e,3,4,1 
finish   
/solu            ! Входим в решатель /SOLUTION 
antype,static    ! Статический тип анализа 
d,all,uz,0       ! Задаем условия закрепления в узлах 
d,2,ux,0 $ d,2,uy,0 
d,4,ux,0 $ d,4,uy,0 
f,3,fy,-100      ! Задаем силу в узле 2+n_fill 
outpr,basic,1    ! Вывод основных результатов в стандартный файл 
solve            ! Решаем СЛАУ МКЭ 
finish  
/exit 

 
Для построения эпюр на криволинейных стержнях можно использо-

вать команду PLLS, аналогично программе SMs3.inp расчета рам из разде-
ла 3. Напомним, что по команде PLLS набор линейных эпюр строится на 
линиях, соединяющих узлы элементов. Поэтому, как и в программе 
SMs3.inp, для более точной картинки эпюр нужно дополнительно сгенери-

ровать узлы и элементы. Чем мельче будут элементы, тем лучше будут вы-
глядеть и получающиеся эпюры. 

Следует также отметить, что при рисовании плоской системы кри-
вых стержней в плоскости Oxy, в соответствии с документацией по эле-
менту PIPE18, поперечные силы в узлах будут находиться в позициях 
SMISC,3 и SMISC,9 (как MFORZ), а изгибающие моменты - в позициях 
SMISC,5 и SMISC,11 (как MMOMY) элементной таблицы ETABLE. Это 
также отличает процедуры рисования эпюр для элементов PIPE18 от рас-
смотренного ранее случая с элементами BEAM3. Соответствующая про-
грамма SMs6.inp, рассчитывающая ту же задачу, что и программа SMs5.inp 
приводится ниже. 
 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                 Файл SMs6.inp 
/com,    ПРОГРАММА РЕШЕНИЯ ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,               ПО ТЕМЕ: "КРИВЫЕ СТЕРЖНИ" 
/com,         (версия 1.1 программы с построением эпюр) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 11.18.    С.331.) 
/com, 
/title, CURVED BAR  
F_R='SMs6' 
/show,F_R,grph   ! Задаем вывод графики в файл <F_R>.grph 
/prep7           ! Входим в препроцессор /PREP7 
                  ! Все данные в системе СГС 
dr=2          ! dr - диаметр стержня 
ar=10         ! ar - радиус кривизны 
et,1,pipe18      ! КЭ - искривленные трубки PIPE18 
r,1,dr,dr/2,ar   ! REAL CONST : 
            !   внешний диаметр сечения OD=dr, 
            !   толщина стенок трубы TKWALL=dr/2 сплошная труба) 
            !   радиус кривизны RADCUR=ar 
mp,ex,1,2e6      ! модуль Юнга EX=2e6 
mp,gxy,1,8e5     ! модуль сдвига GXY=8e5 
n_fill=20        ! число дополнительных узлов на одном КЭ 
csys,1           ! Переходим в цилиндрическую систему координат 
n,1               ! Определяем узлы по координатам 
n,2,ar,180 
n,2+n_fill,ar,90 
fill,2,2+n_fill,n_fill-1  ! Генерируем n_fill-1 равноотстоящих узлов 
                          ! между узлами 2 и 2+n_fill 
n,2+n_fill*2,ar 
fill,2+n_fill,2+n_fill*2,n_fill-1   
                          ! Генерируем n_fill-1 равноотстоящих узлов 

                          ! между узлами 2+n_fill и 2+n_fill*2 
csys,0           ! Возвращаемся в декартову систему координат 
*do,i,2,1+n_fill*2 
  e,i,i+1,1 
*enddo 
finish   
/solu            ! Входим в решатель /SOLUTION 
antype,static    ! Статический тип анализа 
d,all,uz,0        
d,2,ux,0 $ d,2,uy,0  ! Задаем условия закрепления в узлах 
d,2+n_fill*2,ux,0 $ d,2+n_fill*2,uy,0 
f,2+n_fill,fy,-100      ! Задаем силу в узле 2+n_fill 
solve             
finish  
/post1 
 ! Согласно документации по КЭ PIPE18  
 ! определяем указатели на значения нормальных и поперечных сил и 
 ! изгибающих моментов на концах элементов 
etable,mforxi,smisc,1      ! Smisc,1 - нормальная сила N в узле I 
etable,mforxj,smisc,7      ! Smisc,7 - нормальная сила N в узле J 
etable,mforyi,smisc,3      ! Smisc,3 - поперечная сила Q в узле I 
etable,mforyj,smisc,9      ! Smisc,9 - поперечная сила Q в узле J 
etable,mmomzi,smisc,5   ! Smisc,5 - изгибающий момент M в узле I 
etable,mmomzj,smisc,11 ! Smisc,11 - изгибающий момент M в узле J 
/output,F_R,res   ! Направляем вывод в файл <F_R>.res 
prrsol                 ! Печатаем опорные реакции 
pretab,mforyi,mforyj       ! Печатаем поперечные силы 
pretab,mforxi,mforxj       ! Печатаем нормальные силы 
pretab,mmomzi,mmomzj       ! Печатаем изгибающие моменты 
/output 
/pnum,node,1  ! Нумеровать узлы при графическом показе 
eplot         ! Нарисовать элементы 
/pbc,u,,1     ! Показывать граничные условия, силы и моменты 
/pbc,f,,1 
/pbc,m,,1 
/psf,pres,norm,1 ! Показывать распределенные нормальные нагрузки 
/pnum,elem,1  ! Нумеровать элементы 
/pnum,node,0  ! Не нумеровать узлы 
/number,1  ! Занумерованные объекты показывать  
           ! только различным цветом без нумерации 
/zoom,1,scrn,,,1.7 ! Масштабировать окно для удобства 
/pbc,all,,0    ! Не показывать граничные условия, силы и моменты 
/psf,pres,norm,0 ! Не показывать распределенные нагрузки 
/pnum,elem,0     ! Не нумеровать элементы 
/triad,off       ! Не показывать начало и систему координат  
plls,mforyi,mforyj,2  ! Графический показ эпюры поперечных сил Q  
plls,mforxi,mforxj,2  ! Графический показ эпюры нормальных сил N 
plls,mmomzi,mmomzj,-2 ! Графический показ эпюры изгибающих моментов M 
finish  

/exit 
 

 
                Рис. 4.2.                                               Рис. 4.3. 
 

 
                 Рис. 4.4.                                               Рис. 4.5. 
 
 На рис. 4.2–4.5 приведены результаты, полученные по программе 
SMs6.inp. (Реальные графики – цветные.) Рис.4.2 иллюстрирует исходную 
задачу, а на рис.4.3–4.5 приведены соответственно эпюры перерезываю-
щих сил, нормальных сил и изгибающих моментов. Сравнение итоговых 
результатов с ответами, приведенными в [3], убеждает в точности расчетов 
кривых стержней на ANSYS. 

4.3. Несимметричные балки переменного сечения и балки на уп-
ругом основании. Обобщением балочных элементов BEAM3 (2D - пло-
ская балка) и BEAM4 (3D - пространственная балка) на случай балок с не-
симметричными или переменными по длине сечениями являются КЭ 
BEAM54 (2D) и BEAM44 (3D) соответственно. Эти элементы позволяют 
рассчитывать также балки на упругом основании. С использованием КЭ 
BEAM54 и BEAM44 можно решать разнообразные задачи сопротивления 
материалов, как рассмотренные ранее, так и более сложные. Например, 
можно исследовать задачи о косом и пространственном изгибе, внецен-
тренном сжатии, сложном сопротивлении и др. Методики расчетов при 
этом остаются аналогичными приведенным ранее, а самое существенное 
отличие состоит лишь в правильном задании констант (Real Constants) 
элементов, которых, возможно, придется определять значительное число. 
Внимательное изучение документации [6,7] по элементам BEAM54 и 
BEAM44 здесь крайне желательно. Рассмотрим кратко несколько приме-
ров. 

Пусть требуется исследовать изгиб консольной балки, имеющей се-
чение постоянной ширины b и переменной высоты oh x xh l( ) /= . Балка же-
стко защемлена на конце с большей площадью и находится под действием 
равномерно распределенной нагрузки q (Рис. 4.6). Такая задача приведена 
в [3] под номером 8.14. 

 

 
Рис. 4.6. 

 
Для определенности примем следующие числовые данные: материал 

балки - сталь; l=100 (см); h0 10=  (см);  b=5 (см); q=20 (кГ/см). Поскольку 
для балки переменного сечения с распределенной нагрузкой КЭ BEAM54 
не дают точного решения, то для обеспечения приемлемой точности балку 
необходимо дополнительно разбивать на большее число КЭ. Для каждого 
из таких КЭ требуется задавать свой определенный набор констант, по-
скольку размеры балки меняются по длине. Другой сложностью, присущей 
данной задаче, является наличие нулевой площади поперечного сечения на 
левом конце балки, что недопустимо для КЭ BEAM54 ANSYS. Для обхода 
этого препятствия можно задать малую (но отличную от нуля) толщину 
левого поперечного сечения. 

 Программа SMs7.inp, предназначенная для решения данной задачи и 
написанная с учетом изложенного, приводится ниже. 
 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                 Файл SMs7.inp 
/com,    ПРОГРАММА РАСЧЕТА ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,    ПО ТЕМЕ: "ПЛОСКИЙ ИЗГИБ БАЛКИ ПЕРЕМЕННОГО СЕЧЕНИЯ" 
/com,                  (версия 1.1) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 8.14.    С.220-221.) 
/com, 
/com,   Текстовые результаты сохраняются в файле SMs7.res 
/com,   Графические результаты выводятся в файл  SMs7.grph 
/com, 
/prep7           ! Входим в препроцессор /PREP7 
f_r='SMs7' 
/show,f_r,grph   ! В режиме /BATCH задаем вывод графики 
                 ! в файл <f_r>.grph 
antype,static    ! Статический тип анализа 
/title, BENDING OF A TAPERED BEAM 
                 ! Все данные в системе СГС 
h=10           ! Максимальная высота поперечного сечения балки 
b=5            ! Ширина поперечного сечения балки 
lbeam=100      ! Длина балки 
exbeam=2e6     ! Модуль Юнга 
mp,ex,1,exbeam ! MAT. PROP. типа MAT=1: модуль Юнга EX=exbeam 
q=20           ! Значение распределенной нагрузки 
n_el=30        ! Параметр для количества КЭ 
et,1,beam54    ! Элементы с TYPE=1 - , балочные КЭ BEAM54 
hl=lbeam/n_el  ! Длины КЭ 
*dim,x_node,array,n_el+1  ! массив для X-координат узлов 
!  В цикле по k заполняется массив X-координат и создаются узлы 
*do,k,1,n_el+1 
  x_node(k)=hl*(k-1) 
  n,k,x_node(k) 
*enddo 
! Подготовка начальных данных для последующего цикла 
! h1, area1, iz1 - высота, площадь и момент инерции соответственно 
!                  для левого сечения элемента 
h1=h/n_el/10 ! Так как area1 и iz1 не могут равняться нулю, 
             ! то h1 полагается равным в 10 раз меньше следующего 
             ! значения высоты 
area1=b*h1 
iz1=b*h1**3/12 
! В цикле по k для балки с линейно возрастающей толщиной  
! прямоугольного сечения 

! задаются h2, area2 и iz2 для правых сечений КЭ, 
! определяются Real Const. для КЭ, создаются КЭ BEAM54, 
! задается распределенная нагрузка на КЭ 
! и подготавливаются значения h1, area1, iz1 для следующего шага 
*do,k,1,n_el 
 h2=h*x_node(k+1)/lbeam $ area2=b*h2 $  iz2=b*h2**3/12 
 r,k,area1,iz1,h1/2,h1/2,area2,iz2 
 rmore,h2/2,h2/2 
 real,k 
 e,k,k+1 
 sfbeam,k,,pres,20  
 h1=h2 $ area1=area2 $ iz1=iz2 
*enddo 
save 
finish   
/solu            ! Входим в решатель /SOLUTION 
d,n_el+1,ux,,,,,uy,rotz ! Защемление правого края балки 
solve             ! Решаем СЛАУ МКЭ 
finish   
/post1 
 ! Согласно документации по КЭ BEAM54  
 ! определяем указатели на значения поперечных сил и 
 ! изгибающих моментов на концах элементов 
etable,sigmai,ls,2  ! Ls,2 - изгибное напряжение  
                    ! на верхней границе сечения в узле I 
etable,sigmaj,ls,5  ! Ls,5 - аналог. изгибное напряжение в узле J 
etable,mforyi,smisc,2  ! Smisc,2 - поперечная сила Q в узле I 
etable,mforyj,smisc,8  ! Smisc,8 - поперечная сила Q в узле J 
etable,mmomzi,smisc,6  ! Smisc,6 - изгибающий момент M в узле I 
etable,mmomzj,smisc,12 ! Smisc,12 - изгибающий момент M в узле J 
/output,F_R,res   ! Направляем вывод в файл <F_R>.res 
prnsol,u,y   ! Печатаем прогибы балки 
Uymax_ex=-q*lbeam**4/(2*exbeam*iz2) ! Аналитическое значение 
!                                     максимального прогиба балки 
*vwrite,Uymax_ex  
(1x,'   Uymax_ex= ',f14.5) 
prrsol            ! Печатаем опорные реакции 
pretab,mforyi,mforyj       ! Печатаем поперечные силы  
pretab,mmomzi,mmomzj       ! Печатаем изгибающие моменты 
pretab,sigmai,sigmaj       ! Печатаем изгибные напряжения 
/output 
/pnum,elem,1  ! Нумеровать элементы при графическом показе 
/eshape,1     ! Показать сечение балки в 2d - размерности 
eplot         ! Нарисовать элементы 
/eshape,0     ! Не показывать сечение балки в 2d - размерности 
/pbc,u,,1     ! Показывать граничные условия, 
/pbc,f,,1     !                      силы и 
/pbc,m,,1     !                      моменты 
/psf,pres,norm,1 ! Показывать распределенные нормальные нагрузки 

/pnum,elem,1  ! Нумеровать элементы 
/pnum,node,0  ! Не нумеровать узлы 
/number,1  ! Занумерованные объекты показывать  
           ! только различным цветом без нумерации 
eplot 
/pbc,all,,0    ! Не показывать граничные условия, силы и моменты 
/psf,pres,norm,0 ! Не показывать распределенные нагрузки 
/pnum,elem,0     ! Не нумеровать элементы 
/triad,off       ! Не показывать начало и систему координат  
pldisp,1   ! Показать деформированное и недеформированное состояние 
plls,mforyi,mforyj,2  ! Графический показ эпюры поперечных сил Q  
plls,mmomzi,mmomzj,2  ! Графический показ эпюры изгибающих моментов M 
plls,sigmai,sigmaj,2  ! Графический показ эпюры изгибных напряжений 
finish  
/exit 
 
 Получаемые в результате работы программы эпюры поперечных сил, 
изгибающих моментов и максимальных нормальных напряжений здесь не 
приводятся. Отметим, что полученный прогиб на свободном конце получа-
ется равным (-1.2033 см), что практически совпадает с аналитическим зна-
чением, равным (-1.2 см). 
 Расчет плоского изгиба балок на упругом основании может выпол-
няться аналогично. Изменение составляет лишь способ задания констант 
КЭ BEAM54. Входным данным задачи здесь будет являться жесткость уп-
ругого основания EFS, которая должна располагаться в списке Real Con-
stants под номером 16. Для балки постоянной толщины H, площади попе-
речного сечения AREA и момента инерции IZZ константы КЭ BEAM54 
можно определить следующими командами: 
 
R,1,AREA,IZZ,H_2,H_2 
RMODIF,1,16,EFS 
 
где H_2 - половина толщины балки. Команда RMODIF модифицирует те-
кущий список Real Constants, заменяя константу в позиции с номером 16 
(равную по умолчанию нулю) на значение EFS. 
 Следующий пример, взятый из задачи 10.26 [3], демонстрирует рас-
чет внецентренного растяжения-сжатия. Пусть требуется определить мак-
симальные напряжения в стальной полосе с прямоугольным поперечным 
сечением высоты h=18 (см) и толщины b=0.73 (см) при растяжении двумя 
силами P=9000 (кГ), параллельными оси, но приложенными посередине 
толщины на расстоянии δ=5 (см) от края полосы. 

Будем считать, что полоса имеет длину 2l=100 (см). Так как задача 
симметрична относительно среднего сечения полосы, то удобно рассмат-
ривать только половину полосы длины l, а для среднего сечения полосы 
поставить условия симметрии. Для решения такой задачи с внецентренным 

растяжением удобно использовать КЭ BEAM54, хотя возможно и исполь-
зование более общего КЭ BEAM44. На рис. 4.7 показан правый торец по-
лосы вместе ее главными осями инерции, образующими систему коорди-
нат Ox'y'z' . Для КЭ BEAM54 эта система координат является элементной 
системой координат, и положение точки приложения силы (узла конечно-
элементной модели) относительно главных осей инерции задается двумя 
константами: смещениями по осям x' и y'. В нашей задаче смещение по оси 
x' равно нулю, а смещение по оси y' есть d hy = − =/ 2 4δ  (см). Данное 
смещение нужно задавать в списке констант (Real Constants) для КЭ 
BEAM54. Это практически и единственный новый элемент в программе 
SMs8.inp, предназначенной для расчета рассматриваемой задачи. 

 
Рис. 4.7. 

 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                 Файл SMs8.inp 
/com,    ПРОГРАММА РАСЧЕТА ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,    ПО ТЕМЕ: "ВНЕЦЕНТРЕННОЕ РАСТЯЖЕНИЕ-СЖАТИЕ" 
/com,                  (версия 1.1) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 10.26.    С.272.) 
/com, 
/com,   Текстовые результаты сохраняются в файле SMs8.res 
/prep7           ! Входим в препроцессор /PREP7 
f_r='SMs8' 
antype,static    ! Статический тип анализа 
                   ! Все данные в системе СГС 
h=18               ! Высота поперечного сечения балки 
b=0.73            ! Толщина поперечного сечения балки 
lbeam=50       ! Длина половины балки 
exbeam=2e6       ! Модуль Юнга 
mp,ex,1,exbeam ! MAT. PROP. типа MAT=1: модуль Юнга EX=exbeam 
et,1,beam54       ! Элементы с TYPE=1 - , балочные КЭ BEAM54 
n,1                     ! Создаем узлы 

n,2,lbeam 
area1=b*h        ! Площадь поперечного сечения 
iz1=b*h**3/12  ! Момент инерции поперечного сечения 
dy1=4          ! Расстояние по оси Y от узла I до центр. оси 
pf=9e3         ! Значение силы 
r,1,area1,iz1,h/2,h/2,area1,iz1 ! Real Const. с 1 по 6 для балки 
rmore,h/2,h/2,,dy1,,dy1           ! Real Const. с 7 по 12 для балки 
e,1,2           ! Создаем один КЭ 
f,2,fx,pf      ! Прилагаем растягивающую силу 
save 
finish   
/solu            ! Входим в решатель /SOLUTION 
d,1,ux,0 $ d,1,rotz,0  ! Условия, моделирующие симметрию задачи 
solve             ! Решаем СЛАУ МКЭ 
finish   
/post1 
 ! Согласно документации по КЭ BEAM54  
 ! определяем указатели на значения макс. и мин. 
 ! напряжений на концах элемента 
etable,smaxi,nmisc,1   ! Nmisc,1 - макс. напряжение  
                                     ! на сечении в узле I 
etable,smaxj,nmisc,3   ! Nmisc,3 - аналог. напряжение в узле J 
etable,smini,nmisc,2    ! Nmisc,2 - мин. напряжение  
                                     ! на сечении в узле I 
etable,sminj,nmisc,4   ! Nmisc,4 - аналог. напряжение в узле J 
/output,F_R,res           ! Направляем вывод в файл <F_R>.res 
pretab,smaxi,smaxj      ! Печатаем макс. напряжения 
pretab,smini,sminj       ! Печатаем мин. напряжения 
/output 
finish  
/exit 
 
 В результате выполнения данной программы в файле SMs8.res най-
дем значение максимального напряжения в сечение, равное 1598.2 
(кГ/см2 ), что мало отличается от приведенного в [3] к задаче 10.26 значе-
ния [σ]=1600 (кГ/см2 ). 
 В более сложных случаях внецентренного действия силовых факто-
ров, например при отсутствии симметрии точки приложения силы сразу по 
двум осям сечения, следует использовать КЭ BEAM44 (т.е. КЭ балки раз-
мерности 3D). Тем не менее методика, рассмотренная для этого примера, в 
основном сохраняется. 
 
 
 

4.4. Начальные и температурные деформации. Расчет задач с на-
чальными и температурными деформациями и напряжениями не представ-

ляет в ANSYS особого труда. Необходимо отметить только некоторые осо-
бенности. 
 Начальную деформацию для стержневых и балочных КЭ типа LINK 
и BEAM можно задать просто как константу (Real Constant) элемента 
ISTRN. (Эта константа является именно начальной деформацией, то есть 
отношением удлинения к первоначальной длине, а не просто удлинением 
элемента !) Для задач сопротивления материалов с начальными деформа-
циями характерно наличие жесткого элемента конструкции. Жесткая часть 
конструкции не должна деформироваться, и следовательно, не может со-
стоять из стержневых или балочных КЭ. Однако, она накладывает ограни-
чения на перемещения узлов конечно-элементной модели в виде некото-
рых уравнений связей. В ANSYS узловые степени свободы можно подчи-
нить уравнениям связей (constraint equation), используя команду: 
 CE, NEQN,CONST,NODE1,Lab1,C1,NODE2,Lab2,C2,NODE3,Lab3,C3 
Здесь NEQN - ссылочный номер уравнения связи, Lab1, Lab2, Lab3 - обо-
значения степеней свободы в узлах NODE1, NODE2 и NODE3 соответст-
венно, и уравнение имеет вид: 

CONST=Lab1*C1+Lab2*C2+Lab3*C3 
 В качестве примера рассмотрим задачу №1.100 из [3]. В этой задаче 
жесткий брус подвешен на трех стальных стержнях с одинаковыми попе-
речными сечениями площади 20 см2 , расположенными на расстояниях 
ax =100 см друг от друга. Средний стержень короче двух других на δ=0.05 
см, а длины первого и третьего стержней равны ay =100 см. Требуется оп-
ределить напряжения в стержнях после сборки всей конструкции (рис.4.8). 

 
Рис. 4.8. 

 
 Будем считать, что все стержни работают только на растяжение-
сжатие вдоль оси Oy. Тогда отдельный стержень можно моделировать КЭ 
LINK1, а для всех степеней свободы Ux можно принять Ux=0. Зададим уз-
лы и элементы так, как показано на рис. 4.8. Из простых геометрических 
соображений находим, что жесткий брус определяет следующие ограниче-
ния на степени свободы UYj в узлах с j=4,5,6 : 

UY UY5 42− =0 ;     UY UY6 43− =0 
Эти два уравнения и будут являться уравнениями связей между узлами. 
 Для учета начальной деформации второго стержня определим два 
набора констант для элементов LINK1: для первого набора зададим только 
площадь поперечного сечения, а для второго - площадь и начальную де-
формацию. Теперь остается при создании конечных элементов лишь пере-
определить текущие значения номеров констант (Real Constants). В итоге 
приходим к следующей программе. 
 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                 Файл SMs9.inp 
/com,    ПРОГРАММА РАСЧЕТА ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,    ПО ТЕМАМ: "РАСТЯЖЕНИЕ СТЕРЖНЕЙ, НАЧАЛЬНЫЕ 
ДЕФОРМАЦИИ, 
/com,               УРАВНЕНИЯ СВЯЗЕЙ" 
/com,                  (версия 1.1) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 1.100.  С.44.) 
/com, 
/com,   Текстовые результаты сохраняются в файле SMs9.res 
/prep7           ! Входим в препроцессор /PREP7 
f_r='SMs9' 
antype,static    ! Статический тип анализа 
                 ! Все данные в системе СГС 
ax=100 $ ay=100  ! Геометрические размеры  
delt=0.05        ! Начальное удлинение второго стержня 
area=20          ! Площади поперечных сечений 
mp,ex,1,2e6    ! MAT=1: модуль Юнга EX=2e6 
et,1,link1        ! TYPE=1 : стержневой КЭ LINK1 
r,1,area           ! REAL=1 : area 
r,2,area,delt/(ay-delt) ! REAL=2 : area,  
                                    ! ISTRN=delt/(ay-delt) - нач. деформация 
n,1,ax $ n,2,2*ax $ n,3,3*ax ! Создаем узлы 
n,4,ax,-ay $ n,5,2*ax,-ay $ n,6,3*ax,-ay 
! Создаем КЭ с различными Real Const. 
real,1 $ e,1,4 
real,2 $ e,2,5 
real,1 $ e,3,6 
d,all,ux,0    ! Считаем перемещения UX для всех узлов равными 0 
d,1,uy,0,,3   ! Закрепляем верхние узлы (1,2,3) по оси Y 
ce,1,0,5,uy,1,4,uy,-2 ! Задаем 1-ое уравнение связи: UY_5+(-2)*UY_4=0 
ce,2,0,6,uy,1,4,uy,-3 ! Задаем 2-ое уравнение связи: UY_6+(-3)*UY_4=0 
save 
finish   
/solu            ! Входим в решатель /SOLUTION 

solve             ! Решаем СЛАУ 
finish   
/post1 
 ! Согласно документации по КЭ LINK1  
 ! определяем указатели на значения осевых усилий и 
 ! напряжений в стержнях 
etable,forax,smisc,1  ! Smisc,1 - осевое усилие  
etable,sigax,ls,1         ! Ls,1 - осевое напряжение 
/output,F_R,res        ! Направляем вывод в файл <F_R>.res 
pretab,forax,sigax    ! Печатаем усилия и напряжения 
/output 
finish  
/exit 
 
 Результатом работы программы SMs9.inp будут следующие значения 
напряжений в стержнях: σ1= -142.93 (кГ/см2 );  σ2 =714.64 (кГ/см2 );   σ3= 

-428.79 (кГ/см2 ). Соответствующие значения (σ1= -143 (кГ/см2 );  σ2 =715 

(кГ/см2 ); σ3= -429 (кГ/см2 ) ) из ответа к задаче №1.100 [3] практически 
совпадают с числами, полученными на ANSYS. 
 При решении задач с температурными деформациями (напряжения-
ми) необходимо задавать в списке материальных свойств коэффициент 
температурного расширения α , который в ANSYS определяется как ALPX 
по команде: 

MP,NMAT,ALPX,VALUE_ALPX 
где NMAT - номер материальных свойств, VALUE_ALPX - значение коэффи-
циента температурного расширения материала. 
 

 
Рис. 4.9 

 
 Набор команд для определения температурных полей, трактуемых в 
ANSYS как объемные силы (Body Loads), иллюстрируется программой 
SMs10.inp. Эта программа предназначена для решения задачи 1.89в из [3], 
в которой требуется определить наибольшие по абсолютной величине на-

пряжения в стержневой системе (рис. 4.9), подвергающейся нагреванию до 
температуры T=40°C. 

 
 
/batch           ! Пакетный режим /BATCH 
/com, 
/com,                 Файл SMs10.inp 
/com,    ПРОГРАММА РАСЧЕТА ЗАДАЧИ СОПРОТИВЛЕНИЯ МАТЕРИАЛОВ 
/com,    ПО ТЕМАМ: "РАСТЯЖЕНИЕ СТЕРЖНЕЙ, ТЕМПЕРАТУРНЫЕ ДЕФОРМАЦИИ" 
/com,                  (версия 1.1) 
/com, 
/com,   Сборник задач по сопротивлению материалов / 
/com,   Под ред. В.К. Качурина. М.: Наука, 1970. 
/com,            (по задаче 1.89в.  С.40.) 
/com, 
/com,   Текстовые результаты сохраняются в файле SMs10.res 
/prep7           ! Входим в препроцессор /PREP7 
f_r='SMs10' 
antype,static    ! Статический тип анализа 
                 ! Все данные в системе СГС 
ax=50                    ! Геометрический размер  
ar1=20 $ ar2=10   ! Площади поперечных сечений 
mp,ex,1,0.8e6      ! MAT=1: медь 
mp,alpx,1,165e-7 
mp,ex,2,2e6         ! MAT=2: сталь 
mp,alpx,2,125e-7 
et,1,link1             ! TYPE=1 : стержневой КЭ LINK1 
r,1,ar1 $ r,2,ar2  ! REAL 1 и 2 с разными площадями 
n,1,0 $ n,2,ax $ n,3,3*ax !   Создаем узлы 
! Создаем КЭ с различными MP и Real Const.  
mat,1 $ real,1 $ e,1,2 
mat,2 $ real,2 $ e,2,3 
d,all,uy,0           ! Считаем перемещения UY для всех узлов равными 0 
d,1,ux,0,,3,2      ! Закрепляем узлы 1 и 3 по оси X 
! tref,0  ! Так можно задать начальную температуру TREF=0 
             ! (TREF=0 по умолчанию) 
bfunif,temp,40     ! Задаем постоянную температуру T: T-TREF=40 
! Другие способы задания температуры T (T-TREF=40): 
! bf,all,temp,40       !  - для всех узлов  
! bfe,all,temp,1,40  !  - для всех элементов 
finish   
/solu            ! Входим в решатель /SOLUTION 
solve             ! Решаем СЛАУ 
finish   
/post1 
 ! Согласно документации по КЭ LINK1  
 ! определяем указатели на значения осевых усилий и 
 ! напряжений в стержнях 
etable,forax,smisc,1   ! Smisc,1 - осевое усилие  

etable,sigax,ls,1         ! Ls,1 - осевое напряжение 
/output,F_R,res        ! Направляем вывод в файл <F_R>.res 
pretab,forax,sigax    ! Печатаем усилия и напряжения 
/output 
finish  
/exit 
 
 В результате выполнения данной программы получаются следующие 
напряжения в стержнях: σ1= -510.77 (кГ/см2 );  σ2 =-1021.5 (кГ/см2 ). От-

вет к задаче № 1.89в из [3] | |maxσ =1020 (кГ/см2 ) практически совпадает с 
получаемым в ANSYS. 
 
 Варианты задания № 4. Решить задачи из таблицы 4.1 на ANSYS, 
используя данные, приведенные в указанных задачах. 
 

Таблица 4.1. 
 
№  задания 4.1 4.2 4.3 4.4 4.5 4.6 4.7 4.8 4.9 
№ задачи из [3] 11.14 11.15а 11.15б 11.15в 11.15г 11.15д 11.15г 11.16б 11.23 
 
4.10 4.11 4.12 4.13 4.14 4.15 4.16 4.17 4.18 4.19 4.20 
11.24 11.20а 11.20б 11.20в 4.11 4.12 4.35 8.22 8.25 8.27 8.20 
 
4.21 4.22 4.23 4.24 4.25 4.26 4.27 4.28 4.29 4.30 
10.28 8.26 11.25 1.91 1.93 1.101 1.85 11.21 1.91 1.89а 

